
Foreword

No matter how much kids hate school, there are some things
about it that are undeniably fun. Hopefully the programs in this
book will make you think fondly of your own days at school.

School Days is the third of the Itty Bitty Bytes books for the
TI-99/4A® Computer. Each book in the series is designed to
bring you twenty fun programs and, at the same time, teach you
something about BASIC. Typing in the programs is hard work.
You must be careful to number the lines exactly, spell the words
correctly, and put in all the right punctuation. But you will feel
extra good when your program runs smoothly!

Always read the explanations next to each program. Some
times there are additions or changes that you can make to
improve the program. Once you see what the program is all
about, feel free to experiment with it—and by all means com
bine programs that work well together. The more you can do to
make a program reflect your own special personality, the more
you will enjoy it. Let me know if you find some nifty changes!

The Itty Bitty Bytes books have grown out of the teaching
experiences I've had with my computer students. Special thank-
yous go to all my COMPU-KIDS—and to my own kids, too—
who keep bringing me one good idea after another. They are
my severest critics, but my most outstanding asset.

Happy computing!
Claire Bailey Passantino

TI-99/4A® is a registered trademark of Texas Instruments.

A Note to Parents and Teachers
You bought the computer. You read the manuals. You did the demo pro
grams. You and/or your children may even have taken some computer class
es. But now the computer is just sitting there. Everyone was so enthusiastic.
What happened?

The novelty of a new computer willtake you just so far. Beyond that point,
a continuing interest in this incredible tool is directly related to its usefulness.
"Not useful" equates to "not used." So the problem becomes, what can
computers do that children would find useful? What kinds of things would
encourage children to expend the energy needed to create their own comput
er programs?

Each Itty Bitty Bytes book in the Creative Pastimes series is packed with
computer activities appealing to young programmers. Simple games, con
tests, races, pictures, designs, songs, riddles, charts, tests, and more—all are
designed to be fun while reinforcing beginning computer concepts and skills.

Besides providing fun, there are fringe benefits to having children write
their own programs. In working with children, I have found that computer
programming encourages them to:

Think creatively.
Use logical thinking skills.
Attend to details.

Take small steps to achieve a goal.
Personalize programs.
Develop pride and self-esteem.
Appreciate packaged software.

To help children enjoy doing their own programs, here are some sug
gested DOs and DONTs:

• DO encourage children to type in their own programs. With younger
children, bargain: "You type this line and I'll type the next one." (Save the
long lines for yourself!)

• DONT criticize typing expertise. Speed and correct fingering are typing
skills that are minimally related to computer programming.

• DO allow children to make mistakes.

• DO help them find the errors they've made. (This is called "debugging"
the program.)

• DONT worry when there is an error message. This means that a mistake
has been made. Check the program for "bugs."

• DO encourage children to read and understand the program
explanations.

• DON'T, however, force the issue. Some people learn by reading. Others
learn by doing. As skills are repeated over and over in different contexts,
children may just "catch on."

• DO be aware of some common pitfalls. Remember to:
1. Use line numbers.

2. Press ENTER after you type in a line.
3. Give great attention to spelling and punctuation. Quotation marks,

commas, semicolons, colons, and even spaces are very important.
4. Save your program before you turn off the computer.

• DO help children save their programs on tape or disk so they can use
them again and again. If you have a printer, use it to make "hard" copies
of each program. People like to see themselves in print.

• DON'T type the "FCTN" and the plus (+) keys simultaneously. This
combination will erase the program from memory and you will have to
start over.

• DO remember that typing the "FCTN" and the "4" simultaneously will
break the program and get you out of an infinite loop.

• DO type in and run all programs with the ALPHA LOCK on (that is, in the
down position). The only exceptions are the joystick programs, which are
run with the ALPHA LOCK off.

• DO praise children for a job well done. And enjoy the programs that
they've created.

• DO modify and use programs that you yourself find useful.

It is my sincere hope that the Itty Bitty Bytes books will help you and your
children establish a healthy working relationship with your computer. Take
that computer off the shelf! And let me know how things are working out!



School Days
for the TI-99/4A® Computer



Main ideas: Simple graphics
Using CALL KEY for instant input from the keyboard

This is a diagram of the outfit and the key
coordinates used to create it. It may also be
helpful for you to refer to the Character Sets
and the Graphics Sheet in the Appendix. In
this program, character 66 (B) from set 5 is
used to draw the bottom of the outfit. Char

acter 84 (T) from set 7 is used to draw the
top of the outfit. Character 76 (L) from set 6
is used to make the legs by drawing a line
down the middle of the pants. The program
is divided into the following 4 sections.

10 A remark that this section will set things up.
Sets the random numbers and clears the screen.

Makes the screen white (color 16). You may pick a different color.
The slit down the pants will be transparent (color 1). By choosing
transparent, the slit will come out no matter what color you choose
for the screen.

Sets the color of set 5 to light blue (color 6).
Sets the color of set 7 to light red (color 10).

20-30

40

50

60

70

80 A remark.

90-110 Makes 3 rows of character 84, beginning at column 10 and extend
ing for 15 boxes (through box 24).

120-140 Makes 5 rows of character 84, beginning at column 14 and extend
ing for 7 boxes (through box 20).

150 A remark.

160-180 Makes 11 rows of character 66, beginning at column 14 and ex
tending for 7 boxes (through box 20).

190 Makes a vertical column beginning, at row 16, column 17 and ex
tending downward for 7 boxes. The line is made with character 76,
which has been made transparent (in line 50), with all the other
characters in set 6.

200 A remark.

210 Sets up the keyboard for instant input.
220 Ifyou haven't pressed any key, sends you back to line 210 until you

do press a key.
230-240 Picks a color from 1 to 16 for the top (T) and the bottom (B).
250-260 Changes the colors of sets 5 and 7 according to the colors picked in

lines 230 and 240.

270 Goes back to line 210 to wait for you to press another key. Press
FCTN-4 to stop the program.

Clothing the Bod
It's time for school. You can't go naked
Better find something to wear!

10 REM SET THINGS UP

20 RANDOMIZE

30 CALL CLEAR

40 CALL SCREEN(16)

50 CALL C0L0R(6,1,1)
60 CALL C0L0R(5,6,6)
70 CALL COLOR(7,10,10)

80 REM DRAW TOP
90 FOR R=4 TO 6

100 .CALL HCHAR(R,10,84,15)
110 NEXT R

120 FOR R=7 TO 11
130 CALL HCHAR(R,14,84,7)
140 NEXT R

150 REM DRAW BOTTOM

160 FOR R=12 TO 22

170 CALL HCHARfR,14,66,7)
180 NEXT R

190 CALL VCHAR(16,17,76,7)

When you add the last section, you may press any key to
change the color of your outfit.

200 REM CHANGE OUTFITS

210 CALL KEY(0,KEY,STATUS
220 IF STATUS=0 THEN 210
230 T= INT(16*RND)+1
240 B= INT(16*RND)+1
250 CALL C0L0R(5,B,B
260 CALL C0L0R(7,T,T
270 GOTO 210



Main ideas: Writing a computer song

200

205

210

211

212

220-240

250-270

280

290

300

301

310

320

330

350-380

390

Clears the screen.

Makes the screen white (color 16).
Makes the characters in set 2 red (color 9).
Makes the characters in set 3 blue (color 5).
Makes the characters in set 4 black (color 2).
Makes 7 red stripes at every other row beginning at row 3 and
ending at row 15. Each row begins at column 6 and extends for 20
boxes.

Draws a solid blue rectangle by making 7 blue rows beginning at
row 3 and ending at row 9. Each row begins at column 6 but
extends for only 10 boxes, halfway across the flag.
Makes a vertical line for a flagpole.
Skips 9 spaces and prints the word AMERICA.
Reads the note (N) and the time (T) or number of beats to play the
note.

If the number "-1" is read, you skip to line 330.
Plays the note, giving each beat 200 milliseconds. To speed up the
song, choose a lower number. To slow it down, choose a higher
number.

Goes back to line 300 and reads more data.

Ends the program.
The data lines that hold the information for the song. Each line
holds a series of numbers arranged in pairs—note, time, note, time,
note, time, etc.
The flag data referred to in line 301.

This diagram holds all the notes used in the song:



The American Way
School days in America all begin
the same way. Everyone says
the pledge to the flag. Then
comes a patriotic song.
Good morning, boys and girls!

200 CALL CLEAR

205 CALL SCREEN(16)
210 CALL C0L0R(2,9,9)
211 CALL C0L0R(3,5,5)
212 CALL C0L0R(4,2,2)

220 FOR R=3 TO 15 STEP 2

230 CALL HCHAR(R,6,42,20)
240 NEXT R

250 FOR R=3 TO 9

260 CALL HCHAR(R,6,50,10)
270 NEXT R

280 CALL VCHAR(3,6,58,22)
290 PRINT " AMERICA"

300 READ N,T

301 IF N=-l THEN 330

310 CALL SOUND(200*T,N,2)
320 GOTO 300
330 END

350 DATA 262,2,262,2,294,2,247,3,262,1,294,2

355 DATA 330,2,330,2,349,2,330,3,294,1,262,2

360 DATA 294,2,262,2,247,2,262,6

365 DATA 392,2,392,2,392,2,392,3,349,1,330,2

370 DATA 349,2,349,2,349,2,349,3,330,1,294,2
375 DATA 330,2,349,1,330,1,294,1,262,1,330,3,349,1,

392,2

380 DATA 440,1,349,1,330,2,294,2,262,6
390 DATA -1,-1



Main ideas: Inputting variables into a text ^
Using punctuation and spaces in a printout tm
Using FOR...NEXT to repeat an action ^

In this program, the following variables are used. Notice that all the alpha ^
variables (the variables containing letters) have a dollar sign ($). They are
called string variables. N is a numeric variable and has no dollar sign.

TN$ = Teacher's name
N$ = Your name (name of student) ^
S$ = School subject ^
RN$ = Relative's name ^
A$ = Animal
N = Number

PL$ = Place ^
PN$ = Parent's name ^

10 Clears the screen. ^

20-81 Allows you to input the information needed. Notice that the semi- ^
colon (;) makes the input appear right after the prompt. There is no
semicolon at the end of line 40. Did you notice the difference? ^

100 Clears the screen. ^

110-210 Prints the letter. Be careful about quotation marks, semicolons, and
spaces if you want the letter to turn out right.

230-231 Pauses while the computer counts to 3000. This gives you time to ^
read the letter. ^

Guess what happened when the teacher found out the truth! ^

250 PRINT:: ^
260 PRINT "PRESS H TO SEE WHAT ";N$ ^
270 PRINT "HAD TO DO FOR HOMEWORK."
280 CALL KEY(0,K,ST) ^
290 IF ST=0 THEN 280 <*
300 IF K<>72 THEN 280 ^
310 FOR S=l TO 500
320 PRINT S;"I WILL NOT TELL LIES." ^
330 NEXT S "*

How does thisaddition to the program work? First, the computer skips2 lines ^
and prints the directions. Line 280 sets up the keyboard for instant input. As ^
soon as you press H (character 72), you go to line 310. The FOR...NEXT loop
(lines 310 to 330) prints the sentence 500 times. '



The Dog Ate It!
Oh no! You forgot to do your homework. It's a good thing you're
so clever. Here's your homework excuse. You get an A for
creative writing.

10 CALL CLEAR

20 PRINT "TEACHER'S NAME";

21 INPUT TN$

30 PRINT "YOUR NAME";

31 INPUT m
35 PRINT "A SCHOOL SUBJECT";

36 INPUT S$

40 PRINT "NAME OF A RELATIVE"

41 INPUT RN$

50 PRINT "KIND OF ANIMAL";

51 INPUT A$

60 PRINT "A NUMBER";

61 INPUT N

70 PRINT "A PLACE";

71 INPUT PL$

80 PRINT "YOUR PARENT'S NAME";

81 INPUT

100 CALL CLEAR

110 PRINT "DEAR ";TN$;","
120 PRINT N$;" DID NOT DO THE "

125 PRINT S$;" HOMEWORK"

130 PRINT "LAST NIGHT BECAUSE"

140 PRINT RN$;" CAME TO VISIT."

150 PRINT

160 PRINT RN$;"'S ";A$
165 PRINT "GRABBED IT AND TOOK"

170 PRINT N;" BITES OUT OF IT."
180 PRINT "THEN HE RAN AWAY WITH IT TO "

190 PRINT PL$;"."
200 PRINT TAB(12);"YOURS TRULY,"
210 PRINT TAB(12);PN$

230 FOR P=l TO 3000
231 NEXT P

<WW\



Main ideas: A test of homonyms
A test with two choices for each question

10

20

30

40

50

60

70

80

90

100

110

120

130

140

150-170

180

500-540

900

Clears the screen.

Prints the message.
Skips a line.
Reads 3 pieces of information: H$ (the hintfor the homonym),CH$
(the choices), and A$ (the answer).
Checks for the flag data. (If it reads the word END in the H$ posi
tion, it skips to line 150.)
Prints the hint.

Prints the choices. (To make it harder, don't print the choices and
skip this line entirely.)
Allows you to input a guess (G$).
Sends you to line 120 if you guessed correctly.
Tells you the right answer if you guessed incorrectly.
Skips over line 120 to line 130.
Prints "RIGHT!"

Skips a couple of lines.
Goes back to line 40 for the next homonym.
Ends the test.

The official end.

The data lines for the test. To make the test longer, add more data
lines at any line numbers up to line 899. Be sure to follow the same
pattern used in lines 500 to 540.
The flag data (referred to in line 50) that signals the program to end.
The word END is written 3 times because line 40 must read 3

pieces of information.

TALE

01s



Spelling Counts!
Today's spelling lesson is about
homonyms. Eye no watt a
homonym is, dew ewe?

10 CALL CLEAR

20 PRINT "SPELLING COUNTS!"
30 PRINT

40 READ H$,CH$,A$

50 IF H$ = "END" THEN 150

60 PRINT H$

70 PRINT CH$

80 INPUT G$
90 IF G$=A$ THEN 120
100 PRINT "WRONG, IT WAS '

110 GOTO 130

120 PRINT "RIGHT!"

130 PRINT::

140 GOTO 40

150 PRINT

160 PRINT "THIS IS THE END.'

170 PRINT "HOW DID EWE DEW?'

180 END
500 DATA HALLOWEEN QUEEN,(WHICH/WITCH),WITCH
510 DATA EYES DO THIS,(SEE/SEA),SEE
520 DATA A RABBIT,(HARE/HAIR),HARE
530 DATA A GOOD STORY,(TAIL/TALE),TALE
540 DATA PENCILS DO THIS,(RIGHT/WRITE),WRITE
900 DATA END,END,END

HAIR

TO, TOO

WHICH



Main ideas: Drilling arithmetic facts
Using the correct operators (+, -, *, /, =, <,>)

5-10 Randomizes the numbers and clears the screen.

20-30 Picks a number for X (1 to 10) and a number for Y (1 to 10).
40 Counts over 10 spaces and prints the number for X, a plus (+) sign,

the number for Y, and an equal (=) sign. Then it advances the
problem up the screen. (Experiment with the number of colons.)

50 Allows you to input a guess (G).
60 Sends you to line 200 if your guess was correct.
70 Prints this message if your guess was incorrect.
80 Goes back to line 50 so you can guess again.

200 This REM saves the line number for later use.

210 Counts over 20 spaces and prints.
220-230 Pauses while the computer counts to 500.

260 Keeps giving you new problems. Press FCTN-4 to stop.

Flashing your flash cards:

200 A remark to remind you that this is the flashing section.
210 Counts over 20 spaces and prints.

220-250 Makes 20 flashes:

230 Picks a number from 0 to 13, and adds 3 to that number. The result
is a number from 3 to 16. This eliminates color 1 (transparent) and
color 2 (black), which would make it impossible for you to read the
black letters.

239-240 Makes the screen black, and then the color chosen in line 230.
250 Goes back to line 220 for the next flash.

Multiplication:
Note that the X is used to print the problem on the screen, but a star (*) is used
when you want the computer to multiply.

Subtraction:

Line 35 is needed so that your answers are more than or equal to 0. For
example, ifthe computer let X=3 and Y=5, then X-Y= -2. Line 35 eliminates
the possibility of negative answers.

Division:

20 Picks a number from 0 to 100.

30 The +1 eliminates 0, since division by 0 is impossible.
35 Makes the dividend larger than the divisor.
36 Makes sure the answers are integers. For example, 13-^-2 would

be rejected because the answer is not a whole number.
40 Prints the problem.
60 Sends you to line 200 if you gave the right answer.

10



Flashy Flashcards
Mrs. Tiddlebaum's class loves math.

5 RANDOMIZE

10 CALL CLEAR

20 X=INT(11*RND)
30 Y=INT(11*RND)
40 PRINT TAB(10);X;"+";Y;"=
50 INPUT G

60 IF G=X+Y THEN 200

70 PRINT "WRONG! TRY AGAIN!"

80 GOTO 50

200 REM

210 PRINT TAB(20);"RIGHT!";
220 FOR P=l TO 500
230 NEXT P

260 GOTO 10

Her flashcards even flash!

200 REM FLASHY FLASHCARDS
210 PRINT TAB(20);"RIGHT!";
220 FOR F=l TO 20
230 C= INT(14*RND)+3
239 CALL SCREEN(2)

240 CALL SCREEN(C)
250 NEXT F

How about multiplication?

40 PRINT TAB(10);X;"X";Y;"=":
60 IF1 G=X*Y THEN 200

Or subtraction?

35 IF X<Y THEN 20

40 PRINT TAB(10);X;"-";Y;"=":
60 IF G=X-Y THEN 200

Or division?

20 X=INT(101*RND)+ 1
30 Y=INT(10*RND)+1
35 IF X<Y THEN 20

36 IF INT(X/Y)OX/Y THEN 20
40 PRINT TAB(10);X;"/";Y;"=":
60 IF G=X/Y THEN 200

11



Main ideas: Creating a data list
Reading the data to print a selective list

Entering data lines takes a long time, and it's easy to make mistakes. Find a
good friend, and work on this project together. You can check each other as
you go along. Here are some tips:
1. Enter the data one line at a time, beginning at line 400. (Why 400? I'll tell

you later!)
2. Count by 10s when you number the data lines (400, 410, 420, etc.).
3. Remember to start each data line with the word DATA.

4. Put a comma after each piece of data, but not at the end of the line.
5. Always enter the data in the same order: state, abbreviation, capital,

flower, bird. For example:

400 DATA ALABAMA,AL,MONTGOMERY,CAMELLIA,YELL0WHAMMER

6. Run the program after you enter each line to check for errors.

When you finish, you should have a listing of all 50 states. (Ifyou get tired, you
can stop after 10 or 20 states are entered. Add the rest next week when your
typing fingers recover.) Later we will change the program so it does different
things.

Here is an explanation of the basic program:

10 Clears the screen.

40 Reads the data in this order: ST$ (name of the state), AB$ (abbrevi
ation), C$ (capital), F$ (flower), B$ (bird).

50 Checks for the flag data to end the program.
60 Prints the name of the state.

70 Goes back to line 40 to read more data.

400-999 The data lines that hold the information about each state. (Follow
the pattern in line 400.)

1000 The flag data that signals the program to end. Line 40 is waiting to
read 5 words, so write the word END 5 times.

1010 Ends the program.

To see a little of your list at a time, press FCTN-4 to stop the list as it scrolls
up the screen. Then type "CONTINUE" to see a little more.

12



State the States

In social studies you are studying the United States. For extra
credit, you and your best friend have volunteered to enter this
data into your classroom computer. Good luck! (Go on to the
next program to see how to use the data.)

10 CALL CLEAR

40 READ ST$,AB$,C$,F$,B$
50 IF ST$ = "END" THEN 1010
60 PRINT ST$

70 GOTO 40

400 DATA ALABAMA,AL,MONTGOMERY,CAMELLIA,YELLOWHAMMER
I

I

i

1000 DATA END,END,END,END,END
1010 END

(You add the rest of the data here!)

ABBREVIATION

CO

CT

DE

FL
GA
HI

OH

OK

RI

SC

INFORMATION ABOUT THE STATES

STATE

ALABAMA
ALASKA

ARIZONA

ARKANSAS
CALIFORNIA

COLORADO

CONNECTICUT
DELAWARE
FLuRIDA
GEORGIA

HAWAII
IDAHO

ILLINOIS

INDIANA

IOWA

KANSHS

KENTUCKY
LOU1SIHNA

MAINE

MARYLAND

MASSACHUSETTS

MICH1GHN
MINNESOTA
MISSISSIPPI

MISSOURI

MONTANA

NEBRASKA

NEVADA
NEW HAMPSHIRE

NEW JERSEY

NEW MEXICO
NEW YORK

NORTH CAROLINA

NORTH DAKOTA
OHIO
OKLAHOMA

OREGON

PENNSYLVANIA

RHODE ISLAND

SOUTH CAROLINA

SOUTH DAKOTA

TENNESSEE

TEXAS

UTAH
VERMONT

VIRGINIA
WASHINGTON
WEST VIRGINIA

WISCONSIN

WYOMING

CAPITAL

MONTGOMERY

JUNEAU
PHOENIX

LITTLE ROCK
SACRAMENTO

DENVER

HARTFORD
DOVER

TALLAHASSEE
ATLANTA
HONOLULU
BOISE

SPRINGFIELD
INDIANAPOLIS

DES MOINES

TOPEKA

FRANKFORT

BATON ROUGE
AUGUSTA

ANNAPOLIS

BOSTON

LANSING
ST.PAUL
JACKSON

JEFFERSON CITY

HELENA

LINCOLN

CARSON CITY
CONCORD

TRENTON

SANTA FE

ALBANY

RALEIGH

BISMARCK
COLUMBUS
OKLAHOMA CITY

SALEM

HARRISBURG

PROVIDENCE

COLUMBIA

PIERRE

NASHVILLE
AUSTIN

SALT LAKE CITY
MONTPELIER

RICHMOND
OLYMPIA
CHARLESTON
MADISON

CHEYENNE

FLOWER

CAMELLIA

FORGET-ME-NOT
CACTUS BLOSSOM

APPLE BLOSSOM
POPPY

COLUMBINE

MOUNTAIN LAUREL
PEACH BLOSSOM

ORANGE BLOSSOM
ROSE
HIBISCUS
SYRINGA
VIOLET

PEONY
ROSE

SUNFLOWER
GOLDENROD
MAGNOLIA

PINE CONE
BLACKEYED SUSAN

MAYFLOWER
APPLE BLOSSOM

LADY SLIPPER
MAGNOLIA
HAWTHORN

BITTERROOT
GOLDENROD

SAGEBRUSH
LILAC

VIOLET

YUCCA
ROSE

DOGWOOD

ROSE
CARNATION

MISTLETOE

OREGON GRAPE

MOUNTAIN LAUREL

VIOLET

JESSAMINE
PASQUEFLOWER

IRIS
BLUEBONNET

LILY
CLOVER

DOGWOOD
RHODODENDRON
RHODODENDRON

VIOLET

PAINTBRUSH

BIRD

YELLOWHAMMER
GROUSE

CACTUS WREN

MOCKINGBIRD

QUAIL
BUNTING

ROBIN

BLUE HEN
MOCKINGBIRD

THRASHER
GOOSE
BLUEBIRD
CARDINAL

CARDINAL
GOLDFINCH
MEADOWLARK

CARDINAL
PELICAN

CHICKADEE

ORIOLE

CHICKADEE
ROBIN

LOON
MOCKINGBIRD
BLUEBIRD

MEADOWLARK
MEADOWLARK

BLUEBIRD
PURPLE FINCH

GOLDFINCH

ROADRUNNER
BLUEBIRD

CARDINAL

MEADOWLARK
CARDINAL

FLYCATCHER

MEADOWLARK

GROUSE

RED HEN

WREN
PHEASANT

MOCKINGBIRD

MOCKINGBIRD

GULL
THRUSH

CARDINAL
GOLDFINCH
CARDINAL
ROBIN
MEADOWLARK

13



Main ideas: Using IF..THEN to find information on the data list
Using the data list to make a chart
Using the data list to play a game

Use the original program and add these extra lines. As you type the examples,
try to figure out how the new lines fit into the old program. Can you make up
some of your own examples?

To find information:

60 Tells what to look for after each group of data is read. Skips to line
80 if necessary.

80 Prints this message when line 60 is true.
90 Since you skipped line 70, this line goes back to line 40 to read

more data.

To make a chart:

20 Prints the headings.
30 Underlines the headings. (There are several ways to underline.)
60 Tells the computer what to print.

To play a game:

60-61 Prints a question for you to answer.
62 Allows you to input your guess (A$).
63 Sends you to line 80 if your guess was correct.
64 Prints this message if your guess was incorrect.
80 Prints "RIGHT!" if your guess was correct.
90 Ifyour guess was correct, you skipped line 70. You need line 90 to

go back to line 40 to read more data.

To score the game:

85 Adds 1 to the number correct (C). Bydefault, when you type "RUN"
to begin the game, C is set to 0.

1010 Prints your score just before the program ends.
1020 Ends the program.

Trivia Expert
Let the computer turn you into a trivia expert. Who else could
possibly know the state bird of Wyoming? Now that you put in
all that data, let the computer help you:

To find information: or

60 IF ST$="WYOMING" THEN 80 60 IF AB$="CT" THEN 80
80 PRINT B$

90 GOTO 40

or

60 IF F$ = "VIOLET" THEN 80

80 PRINT ST$

90 GOTO 40

To make a chart:

20 PRINT "STATE","CAPITAL"

30 PRINT " "," "
60 PRINT ST$,C$

or

20 PRINT "STATE","FLOWER"

30 PRINT "*****","******"

60 PRINT ST$,F$

To play a game:

60 PRINT "THE STATE IS ";ST$
61 PRINT "WHAT'S THE ABBREVIATION";

62 INPUT A$

63 IF A$=AB$ THEN 80

64 PRINT "WRONG, IT WAS ";AB$::

80 PRINT "RIGHT!"::

90 GOTO 40

To score the game:

85 C=C+ 1

1010 PRINT "YOUR SCORE IS ";C*2;"%"
1020 END

80 PRINT C$
90 GOTO 40

or

60 IF B$ = "CARDINAL" THEN 80
80 PRINT ST$

90 GOTO 40

or

20 PRINT "STATE","BIRD"
30 PRINT "=====","===="

60 PRINT AB$,B$

15

Main ideas: A race across the screen ^
Using ON...GOTO

10 Clears the screen. ^
20-50 Allows you to input your choices for the 3 worst lunches.

60 Clears the screen again. ^
70-100 Sets up the screen. Each of these lines prints something and then «**

advances it up the screen by using colons.
110-130 Although the screen is 32 boxes across, the text begins at box 3.

Therefore, all the lunches begin their race at box 3.
140 Ensures a random outcome. /^

150 Picks a number from 1 to 3 (a vote for the worst lunch).
160 The ON...GOTO statement saves you some IF..THEN statements.

Ifthe computer picked a 1, it goes to line 200. Ifit picked a 2, it goes
to line 300. If it picked a 3, it goes to line 400.

200-230 This is what happens if the computer picked a 1:
200 Prints # (character 35) at box A right below the first lunch.
210 Adds 1 to A so the next # is printed next to the last #.
220 Checks to see if A is 33. Since the screen goes only to 32, if any <**

lunch gets to 33, it must be a winner. If A=33, you go to line 500. ^
230 Otherwise, you go back to line 150. The computer picks the next

lunch to get a vote.
300-330 This is what happens if the computer picked a 2. ^
400-430 This is what happens if the computer picked a 3. ^

500 Prints this message when there is a winner—or should I say ^
"loser"?

fff\

/tfP\

rfSfix.

MS.

Gag Me with a Spoon
It's lunch time at last, but—oh no! It's the tuna salad boat again!
What do you think is the worst school lunch?

10 CALL CLEAR
20 PRINT "THE 3 WORST LUNCHES ARE:"

30 INPUT A$
40 INPUT B$

50 INPUT C$

60 CALL CLEAR
70 PRINT " THE WORST SCHOOL LUNCH"::::

80 PRINT A$:::::

90 PRINT B$:::::
100 PRINT C$::::
110 A=3

120 B=3

130 C=3
140 RANDOMIZE

150 X=INT(3*RND)+1
160 ON X GOTO 200,300,400

200 CALL HCHAR(15,A,35)

210 A=A+1 J83^fJf//^\ A
220 IF A=33 THEN 500 T^lJJHb I \ W
230 GOTO 150
300 CALL HCHAR(20,B,64) T"f "^.^
310 B=B+1 XTxVv^y
320 IF B=33 THEN 500 /S^ V*#*
330 GOTO 150 /^ Jc,
400 CALL HCHAR(25,C,42)
410 C=C + 1
420 IF C=33 THEN 500
430 GOTO 150
500 PRINT "LET'S DROP IT FROM THE MENU!" r*\ &^P

Main ideas: Inserting variables into a text
Using CALL KEYfor "Press any key"

10 Clears the screen.

20-110 Allows you to input your ideas.
120 Clears the screen.

150-180 Prints the poem. Be careful with the spaces, semicolons, and
colons. Remember that the semicolon (;) signals the computer to
print a word right next to the last word printed. The colon (:) makes
the computer skip a line.

190 By making the computer skip lines, we move the poem up to the
middle of the screen.

200 Prints the direction.

210 Sets up the keyboard for instant input.
220 If you haven't pressed any key, you keep going back to line 210

until you do.
230 As soon as any key is pressed, you arrive at this line number, which

sends you back to the beginning at line 10.

This section makes each verse appear in a different color:

130 Picks a number from 1 to 16 for the color (C).
131-132 Screens that are transparent (color 1) or black (color 2) are dis

qualified because the black letters will not show up on these colors.
140 Otherwise, the screen becomes the color chosen.

18



Jump Rope
After lunch the girls like to jump rope. (I guess the boys don't
know how!) This is one of their favorite poems. Why not make
up some verses of your own?

10 CALL CLEAR
20 PRINT "LETTER OF ALPHABET";

30 INPUT L$
40 PRINT "GIRL'S NAME STARTING WITH "; L$
50 INPUT G$
60 PRINT "BOY'S NAME STARTING WITH ";L$
70 INPUT B$
80 PRINT "A PLACE THAT STARTS WITH ";L$
90 INPUT PL$
100 PRINT "SOMETHING THAT BEGINS WITH ";L$;" (A

PLURAL)";

110 INPUT TH$

120 CALL CLEAR
150 PRINT L$;",MY NAME IS ";G$;" AND"::
160 PRINT "MY HUSBAND'S NAME IS ";B$::
170 PRINT "WE COME FROM ";PL$::
180 PRINT "JUST TO SELL YOU ";TH$
190 PRINT ::::::

200 PRINT "PRESS ANY KEY TO CONTINUE."
210 CALL KEY(0,N,STATUS)
220 IF STATUS =0 THEN 210
230 GOTO 10

130 C= INT(16*RND)+1
131 IF C=l THEN 130

132 IF C=2 THEN 130

140 CALL SCREEN(C)

Main ideas: Using CALL KEY for instant input from the keyboard
Drawing with the arrow keys

10 Clears the screen.

20 Sets the color of your path to blue (color 5). Use a different number
from 3 to 16 if you like.

30 Randomizes the numbers. Otherwise, Joe's keys are located at the
same place again and again.

40-50 Selects a hidden row (HR) and a hidden column (HC) for the loca
tion of the missing keys.

60-70 Sets your position (row and column) right in the middle of the
screen.

80 Prints the message.
90 Prints your position (a blue box) on the screen.

100 Ifyour row is the same as the hidden row, the program skips to line
250 to see if you're at the right column or not.

110 Sets up the keyboard for instant input.
120 If you haven't pushed any key, sends you back to line 110 untilyou

do!

130-160 Checks to see which key you have pressed:
Character 69 is E, where the up arrow is located.
Character 88 is X, where the down arrow is located.
Character 68 is D, where the right arrow is located.
Character 83 is S, where the left arrow is located.

170-240 Adds 1 to, or subtracts 1 from, the row or the column of your
position on the screen, depending on which arrow key you pressed.

250 If your row is the same as the hidden row, you were sent to this line.
If your column is different from the hidden column, you are sent
back to line 110.

260 Otherwise, by deduction, you have found both the hidden row and
the hidden column. Hooray for you!

270 Sets the program into an infinite loop for an ending. Press FCTN-4
to break out of the program.

175-246 If you try to plot a point that is not on the screen, you get an error
message. Reason your way through this section to see how it
keeps you from going off the screen.

20



Joe's Keys
Joe, the custodian, is crawling around

on his hands and knees. He has lost

his keys somewhere. Poor Joe needs
all the help he can get. Why don't you

help Joe find his keys?

10 CALL CLEAR

20 CALL C0L0R(2,5,5)
30 RANDOMIZE
40 HR=INT(24*RND)+1
50 HC = INT(32*RND)+1

60 R=12

70 C= 16
80 PRINT "HELP JOE FIND HIS KEYS! ii

90 CALL HCHAR(R,C,42)
100 IF R=HR THEN 250
110 CALL KEY(0,KEY,STATUS) <
120 IF STATUS=0 THEN 110
130 IF KEY=69 THEN 170
140 IF KEY=88 THEN 190
150 IF KEY=68 THEN 210
160 IF KEY=83 THEN 230 '
170 R=R-1

180 GOTO 90

190 R=R+1

200 GOTO 90

210 C= C+ 1

220 GOTO 90
230 C=C-1 175 IF R<1 THEN 185

240 GOTO 90 185 R=l

250 IF COHC THEN 110 186 GOTO 90

260 PRINT "YOU FOUND THEM!!" 195 IF R>24 THEN 205

270 GOTO 260 205 R=24

206 GOTO 90

215 IF C>32 THEN 225

225 C=32

226 GOTO 90

235 IF C<1 THEN 245

245 C = l

246 GOTO 90

Main ideas: A multiple-choice test

10 Clears the screen.

20-30 Prints the title and underlines it. TAB makes it count over to the 5th

column before it starts to print.
40 Skips a line.

50-100 Each line prints 5 spaces and the name of an animal group.
110 Skips a line.
120 Prints the directions.

130 Skips a line.
140 Presents the questions, 3 at a time. This is how many questions fit

on the screen if the choices are to remain visible.

150 Reads the name of an animal (AN$) and its classification number
(CL). (Notice that the animal variable is a string variable and has a
dollar sign ($), while the class variable is a numeric variable and
has no dollar sign.)

160 When the word END is read, you are sent to line 280.
170 Skips a line.
180 Asks the question.
190 Allows you to input your guess (G).
200 Sends you to line 230 if your guess was correct (if G = CL).
210 Prints this message if your guess was incorrect and tells you the

correct answer.

220 Skips to line 240 to avoid the "RIGHT!" message in line 230.
230 Prints "RIGHT!"

240 Keeps going back to line 140 until 3 questions are presented.
250-260 Pauses while the computer counts to 300. (Leave out these lines to

see why you need them.)
270 Goes back to line 10 each time 3 questions have been asked and

answered.

280 Ends the program.
300-340 The data lines that hold the information for the test questions.

Notice that they are arranged in the same pattern. (If you want to
make the test longer, you may use lines 341 through 399 to add
more data. If you choose to add more data, be sure to follow the
pattern in lines 300 to 340.)

400 The flag data that signals the program to end (see line 160).

Science

Lunch time is over.

The bell rings, and it's back to
class you go. It's time for science.
Did you know that all animals
belong to one of these six groups?

10 CALL CLEAR
20 PRINT TAB(5) ;"CLASSES OF ANIMALS

30 PRINT TAB(5) . ii

'

40 PRINT

50 PRINT " 1.INVERTEBRATES"

60 PRINT " 2.FISH"

70 PRINT " 3.AMPHIBIANS"

80 PRINT " 4.REPTILES"

90 PRINT " 5.BIRDS" i

100 PRINT " 6.MAMMALS" ^
110 PRINT

120 PRINT "PICK A NUMBER (1-6)";
130 PRINT

140 FOR Q=l TO 3
150 READ AN$,CL
160 IF AN$= "END" THEN 280
170 PRINT

180 PRINT "WHAT IS A ";AN$;

190 INPUT G

200 IF G=CL THEN 230
210 PRINT "WRONG,IT BELONGS TO GROUP

220 GOTO 240
230 PRINT "RIGHT!"

240 NEXT Q
250 FOR P=l TO 300
260 NEXT P

270 GOTO 10

280 END
300 DATA DOG,6,TURTLE,4,STARFISH,1
310 DATA WHALE,6,TURKEY,5,FROG,3

320 DATA SALAMANDER,3,LIZARD,4,PEACOCK,5

330 DATA BAT,6,CLAM,1,SPIDER,1
340 DATA PERSON,6,HIPPOPOTAMUS,6,SALMON,2

400 DATA END,0

1;CL

Main ideas: Using the joystick to draw on the screen
Using CALL KEY to change colors

To help you understand this program, first type in:

1 CALL KEY(0,KEY,STATUS)
2 PRINT KEY

3 GOTO 1 "'" o*-4

Find the Character Sets in the Appendix. Run the program with the ALPHA
LOCK on, the way we usually do. You will get a string of "-1" untilyou press a
key. When you press a key, its code appears on the screen. Can you figure
out why you get a different number with the ALPHA LOCK off? (A small A has
a different code from a big A!)

For the joystick to operate correctly, the ALPHA LOCK must come off before
you run the program. During the program, then, we use the code numbers for
small R (character 114 in set 11), small Y (character 121 in set 12), and small
B (character 98 in set 9). Refer to lines 30, 40, and 50.

10-20 Clears the screen and makes it white (color 16).
30 Makes the characters in set 9 blue on blue (blue squares).
40 Makes the characters in set 11 red on red (red squares).
50 Makes the characters in set 12 yellow on yellow (yellow squares).

60-70 Sets the beginning row and column numbers at midscreen.
80 Sets the initial character at 114, the code for small R. This means

that when you start the program, the computer draws in red.
90 Checks to see if you've pressed a key.

100 Goes to line 420.

110 Reads the joystick coordinates.
120-230 Adds 1 to or subtracts 1 from the row number (R) and/or the col

umn number (C). This depends on which direction you pushed the
joystick. Refer to the diagram and see ifyou can figure out how this
section works.

300 A remark to save this line. Later we'll add lines 300 to 360. These

lines will keep us on the screen by keeping the column number
between 1 and 32 and the row number between 1 and 24.

400 Plots a red, blue, or yellow square at the current row and column.
The color of the square depends on the character number.

410 Goes back to line 90 to draw another square.
420-450 The GOSUB routine for line 100:

430 If you didn't press a key, it skips to line 450.
440 Ifyou pressed a key, the code for the key becomes the number for

the character to be drawn.

450 Returns to the main program.

Note: When you press FCTN-4 to break out of the program, the colors
disappear and you will see all the little Rs, Ys, and Bs.

Modern Art

The art teacher is absent, and

the sub lets you draw whatever you like
Press R for red, Y for yellow, and
B for blue. Maybe you can exhibit
your work at the Modern
Art Museum!

10 CALL CLEAR

20 CALL SCREEN(16)
30 CALL C0L0R(9,5,5)

40 CALL C0L0R(11,9,9)
50 CALL C0L0R(12,12,12)
60 C= 16
70 R=12

80 CHAR=114

90 CALL KEY(0,KEY,STATUS)
100 GOSUB 420

110 CALL J0YST(1,X,Y)
120 IF X=-4 THEN 150
130 IF X=4 THEN 170
140 GOTO 180

150 C=C-1

160 GOTO 180

170 C= C+ 1
180 IF Y=-4 THEN 210
190 IF Y=4 THEN 230

200 GOTO 300
210 R=R+1

220 GOTO 300

230 R=R-1

300 REM
400 CALL HCHAR(R,C,CHAR)

410 GOTO 90

420 REM CHANGES COLOR

IF NECESSARY

430 IF KEY=-1 THEN 450
440 LET CHAR=KEY

450 RETURN

300 REM KEEP ON CANVAS
310 IF R>24 THEN 320
311 IF R<1 THEN 330

312 GOTO 340

320 R=24

321 GOTO 340

330 R=l
340 IF C>32 THEN 350
341 IF C<1 THEN 360

342 GOTO 400

350 C=32
351 GOTO 400

360 C= l

Main ideas: Using CALL KEY for instant inputfrom the keyboard ASCII codes
Converting keystrokes to musical notes

10 Clears the screen.

20 Prints the direction. The colons advance it up the screen.
30 Sets up the keyboard for instant input. As soon as you press one of

the desired keys, a note will play.
40 If you haven't pressed any key, you keep going back to line 30 until

you do!
41 If you leave out this line, the note may play repeatedly when you

wanted to hear it only once. These repetitions occur because you
don't take your hand off the key fast enough. (Leave out this line
and see what I mean!) The "-1" status means that the same key
has been pressed.

50 Refer to the Character Sets in the Appendix. You will see that
character 49 is 1, character 50 is 2, and so forth. If you press a 1,
you are sent to line 150, and the note becomes 262 (middle C).
Then, in line 151, you are sent to line 300, which plays the note.

60-211 See ifyou can trace this section of the program. It is responsible for
assigning notes to each key pressed. The octave used in the pro
gram is illustrated below. For a more complete keyboard, refer to
the Musical Notes in the Appendix.

m ITT i
262 294 330 349 392 440 494 523

1 3 4 6 7 8

300 Plays each note for 300 milliseconds.
310 Goes to line 30 to see what you press next.

These lines change the background color for each note played:

305 Picks a number from 1 to 16 for the color (C).
306-307 Transparent (color 1) and black (color 2) are disqualified because

the print will not show up.
308 Makes the screen the color chosen.

Music

The music teacher has asked everyone to compose a tune. It's
time to play your song. Do a good job!

10 CALL CLEAR

20 PRINT "TO MAKE MUSIC PRESS KEYS 1-8"::::::::::

30 CALL KEY(0,K,STATUS)
40 IF STATUS=0 THEN 30
41 IF STATUS=-1 THEN 30

50 IF K=49 THEN 150

60 IF K=50 THEN 160

70 IF K=51 THEN 170
80 IF K=52 THEN 180
90 IF K=53 THEN 190
100 IF K=54 THEN 200
110 IF K=55 THEN 210

120 N=523

121 GOTO 300

150 N=262

151 GOTO 300
160 N=294

161 GOTO 300

170 N=330

171 GOTO 300

180 N=349

181 GOTO 300

190 N=392

191 GOTO 300

200 N=440

201 GOTO 300

210 N=494

211 GOTO 300

300 CALL SOUND(300,N,2)
310 GOTO 30

You'll get extra credit
for an interesting performance.

305 C= INT(16*RND)+1
306 IF C=l THEN 305

307 IF C=2 THEN 305

308 CALL SCREEN(C)

Main ideas: Averaging a list of numbers
Using CALL KEY for "Press any key"

10 Clears the screen.

20-30 Allows you to input the student's name.
40 Skips a line.

50-60 Prints the directions.

70 The total number of tests (T) is set to 0.
80 The sum of the test scores (S) is set at 0.
90 Allows you to input a test score (SC).

100 If you type a "-1," you are sent to line 140 immediately.
110 The number of tests goes up 1.
120 The sum of the scores (S) is increased by whatever number the

score (SC) was. For example, if the sum is 0 and you score 90 on
the first test, the new sum is now 90. Ifyou score 90 on the second
test, the sum becomes 180, and so forth.

130 Goes back to line 90 so you can enter the next test score.
140 A remark.

150 Computes the average. The average (AV) is the sum of the scores
(S), divided by the number of tests (T). For example, if the sum of
all the test scores is 450 and you took 5 tests, your average is 90.

160 Prints the average.
170-190 Evaluates your average, and gives you a grade by sending you to

an appropriate line number.
200 If your score was below 70, you are automatically given a U (for

unsatisfactory).
210-270 Prints your grade and sends you to line 280.

280 Ends the program

This program works just fine for a single student, but if you have several
students, you might liketo add these lines so you don't have to keep typing the
word RUN.

280 Skips a line.
290 Prints the direction.

300 Sets up the keyboard for instant input.
310 If you haven't pressed any key, you keep going back to line 300

until you do!
320 As soon as you press a key, you go back to line 10 to start over with

a new student.

If you add lines 280 to 320, you will create an "infinite loop" situation. You
must type FCTN-4 to break out of the program.

Report Card
Today is report card day. Average your test grades, and find out
what you got on your report card.

10 CALL CLEAR

20 PRINT "NAME OF STUDENT:"

30 INPUT N$
40 PRINT

50 PRINT "SCORES ON TESTS:"

60 PRINT "(TYPE -1 WHEN FINISHED)"
70 T=0
80 S= 0

90 INPUT SC

100 IF SC=-1 THEN 140
110 T=T+1

120 S=S+SC

130 GOTO 90
140 REM FIND AVERAGE

150 AV=S/T
160 PRINT "THE AVERAGE IS ";AV
170 IF AV>=90 THEN 220
180 IF AV>=80 THEN 240
190 IF AV>=70 THEN 260
200 PRINT N$;" GETS A U!"
210 GOTO 280

220 PRINT N$;" GETS AN A!"
230 GOTO 280
240 PRINT N$;" GETS A B!"
250 GOTO 280

260 PRINT N$;" GETS AC!"
270 GOTO 280
280 END

280 PRINT

290 PRINT "PRESS ANY KEY TO CONTINUE."

300 CALL KEY(0,N,STATUS)
310 IF STATUS= 0 THEN 300
320 GOTO 10

Main Ideas: Scrolling
Using the joystick

When you type something on the screen, everything previously typed scrolls
up the screen. If there is something printed at the top of the screen, it scrolls
right off the screen. This program takes advantage of the scrolling action.

10 Clears the screen.

20 The position of the bicycle (B) is set at 16, which is halfway across
the screen.

30 Character 50 in set 3 is being used for the bicycle (really a square).
40 The number of loops this program makes is set to 100.
50 Picks a number from 1 to 32 for the position of the car (C).
60 Picks a number from 1 to 32 for the position of the truck (T).
70 Reads the joystick numbers for X and Y.
80 Goes to line 175 to interpret the joystick reading.
90 Prints a colon (:) for the truck at row 23 near the bottom of the

screen.

100 Prints an @ symbol for the car at row 24, the last row on the screen.
(You can change the symbols for the car and the truck by referring
to the Character Sets in the Appendix and changing the character
numbers.)

110 Draws a bicycle about halfway up the screen at row 12.
120 This line is very important. It prints a blank line so the whole screen

scrolls up.
130 Goes back to line 40 for the next loop. [:jf4.

140-170 Prints an ending; ends the program.
175 A reminder that this section reads the

30

joystick.
180 If you push the joystick to the left, ^ .

the Xcoordinate is -4 (see the diagram). '""* 0*.4
You go to line 210.

190 If you push the joystick to the right, the X coordinate is 4 (see the
diagram). You go to line 260.

210 Subtracts 1 from the position of the bicycle.
220 Checks to see if you're about to go off the screen.
230 If you're not, you go to line 300.
240 Ifyou are, the bicycle is kept at position 1 and line 250 sends you to

line 300.

260 Adds 1 to the position of the bicycle.
270 Ifyou're about to go off the screen, the bicycle is kept at position 32

(line 290), and finally you are sent to line 300.
300 Returns to the main part of the program to put the bicycle at the

proper position on the screen.



The Ride Home

Your Mom is really going to like your report card. She'll probably
make your favorite after-school treat. You decide to race home
on your bicycle. Sometimes you even race against traffic.
Watch out!

10 CALL CLEAR

20 B= 16
30 CALL C0L0R(3,5,5)
40 FOR N=l TO 100
50 C=INT(32*RND)+1
60 T=INT(32*RND)+1
70 CALL J0YST(1,X,Y)
80 GOSUB 175

90 CALL HCHAR(23,T,58)
100 CALL HCHAR(24,C,64)
110 CALL HCHAR(12,B,50)
120 PRINT

130 NEXT N
140 PRINT "YOU'RE HOME!"*
150 PRINT "HOW MANY FENDER BENDERS"

160 PRINT " DID YOU HAVE?"

170 END
175 REM READING THE JOYSTICK

180 IF X=-4 THEN 210
190 IF X=4 THEN 260
200 GOTO 300
210 B=B-1

220 IF B<1 THEN 240

230 GOTO 300

240 B= l

250 GOTO 300
260 B=B+ 1

270 IF B>32 THEN 290
280 GOTO 300

290 B=32
300 RETURN

Main ideas: Counting backwards
Using flag data
RESTORE-ing data to be used again

10 Begins a FOR...NEXT loopthat will count backwards from 99 to 1.
(The end of the loop is at line 120.)

20 Clears the screen.

30-40 Prints the first line of the song.
50 Reads the note (N) and the time (T) that the note gets. The time is

given as number of beats.
60-90 Each line here checks to see if flag data has been read. Look at

lines 280 to 310, noticing the data at the end of each line. Each
number signals the computer to go to a certain lineof the program,
which directs it to print the next line of the song on the screen.

100 Plays the note for 100 milliseconds per beat. (Less than 100 milli
seconds, the song goes faster; more than 100, it goes slower.)

110 Goes back to line 50 to read the next note.

120 Finishes the loop, and sends you back for more.
130-200 On cue the computer prints the lines of the song, one at a time.

Then it goes back to line 50 to read more notes.
210 Restores the data so it can be read again for the next verse of the

song.

220 Advances the verse up the screen.
230 Ifthere is more than one bottle left on that wall, you go back to line

120 for another verse.

240-260 Prints these lines when the song is over.
270 A "dummy"line is used to freeze the screen. (TypeFCTN-4 to get

out of this infinite loop.)
280-310 The data lines that hold the information for the program. (Refer to

the Musical Notes in the Appendix if you want to see where the
numbers come from.)

This little addition makes the screen change color for each verse.

21 Randomizes the numbers.

22 Picks a number from 1 to 16 for the color.
23-24 If transparent (color 1) or black (color 2) is chosen, you are sent

back to line 22 for another color selection. (The letters will not show
up on a black or transparent screen.)

25 Makes the screen the color chosen.

Field Trip
Today's the day of the big game. Your school is playing an arch
rival. The school has provided busses to get to the game. On
the bus, the kids sing this old classic.

10 FOR B=99 TO 1 STEP-1

20 CALL CLEAR

30 PRINT B;"BOTTLES OF MILK"

40 PRINT " ON THE WALL": : /^s^n/Tlflfl
50 READ N,T J/jU^^^^
60 IF N=-l THEN 130

70 IF N=-2 THEN 150 Kja\ ^f^l
80 IF N=-3 THEN 180
90 IF N=-4 THEN 210
100 CALL SOUND(100*T,N,2)
110 GOTO 50
120 NEXT B

130 PRINT B;"BOTTLES OF MILK-"::
140 GOTO 50
150 PRINT " TAKE ONE DOWN AND"

160 PRINT " PASS IT AROUND"::

170 GOTO 50

180 PRINT B-l;"BOTTLES OF MILK"

190 PRINT " ON THE WALL!"

200 GOTO 50

210 RESTORE

220 PRINT ::::::

230 IF B>1 THEN 120

240 PRINT "HOW CAN THE BUSDRIVER DRIVE"

250 PRINT "WITH HIS FINGERS IN HIS EARS"

260 PRINT "????????????????????????????"

270 GOTO 270

280 DATA 349,1,349,1,349,1,349,1,

349,1,349,1,349,1,349,1,349,1, why not change
349,3,-1,-1 the color for

290 DATA 392,1,392,1,392,1,294,1, . ?
294,1,294,1,392,6,-2,-2

300 DATA 330,3,330,1,330,3,330,1, 21 RANDOMIZE
330,1,330,1,330,1,330,3,-3,-3 22 C=INT(16*RND) +1

310 DATA 262,1,262,1,262,1,294,1, 23 IF C=1 THEN 22
294,1,330,1,349,1,349,1,349,1, 24 IF C=2 THEN 22
349,3,-4 -4 25 CALL SCREEN(C)

Main ideas: Random outcome of a game
Incrementing a total

10 Clears the screen.

20-50 Allows you to input the team names.
60 Clears the screen.

70 Makes the screen cyan (color 8). (Refer to the Colors in the Appen
dix to select a different color.)

80-90 Counts over to the 5th column and prints the title.
100 Skips a line.

110-120 Atthe beginning of the game, yourteam (Y) has no points and their
team (TH) has no points.

130 The game keeps comingout the same way unless you use this line.
140 Since there's a basket a minute during the game, this loop will

repeat 60 times.
150 Picks a number for the basket (B), a 0 or a 1.
160 If it picked a 0, you skip to line 190 where your team gets 2 points

added to its total.

170 Otherwise, their team gets the 2 points and line 180 sends you to
line 200.

200 Prints the score. Notice the difference between the variable names.
Y$ stands for the name of the team; it is a string variable. Y stands
for the number of points your team scored; it is a numeric variable
and has no dollar sign ($). (The same applies to TH$, which is a
string variable, and TH, which is a numeric variable.)

210 Goes back to line 140 for the next "minute" of the game.
220 Skips a line.
230 If the score is tied, you are sent to line 260.
240 If the score is uneven, somebody won, and this line is printed.
250 Sends you to the end of the program at line 270.
260 Prints this message when the score is tied.
270 Ends the program.

The Big Game *

What a game!
There was a basket scored every minute.
Here's how the game turned out:

10 CALL CLEAR

20 PRINT "NAME OF YOUR TEAM";

30 INPUT Y$
40 PRINT "NAME OF THEIR TEAM";

50 INPUT TH$

60 CALL CLEAR

70 CALL SCREEN(8)
80 PRINT TAB(5);"THE BIG GAME"

90 PRINT TAB(5);"
100 PRINT

110 Y=0
120 TH=0

130 RANDOMIZE

140 FOR M=l TO 60
150 B= INT(2*RND)

160 IF B=0 THEN 190
170 TH=TH+2

180 GOTO 200

190 Y=Y+2

200 PRINT Y$;Y,TH$;TH

210 NEXT M

220 PRINT

230 IF Y=TH THEN 260
240 PRINT TAB(10);"FINAL!!"
250 GOTO 270
260 PRINT TAB(10);"REMATCH!"
270 END

35



Main ideas: Making a chart
Using TAB, commas, and semicolons to format printouts
Using formulas to compute results

The TAB command counts over to a certain column on the screen before

printing. The semicolon prints one piece of information right next to another.
The comma sends the cursor halfway across the screen to the next print zone.
These three devices were used throughout this program.

10 Clears the screen.

20-50 Allows you to choose your pizza and tell how much it costs.
60 Clears the screen again.
70 Counts over to column 5 and prints the kind of pizza (K$).
80 Counts over to column 3 and underlines the kind of pizza.
90 Prints the price of the pizza (P).

100 Figures the tip (TP) as 15% of the price.
110 Prints the tip at column 14, next to the price.
120 Figures the new price as the old price plus the tip.
130 Prints the total price (including the tip).
140 Skips a line.

150-180 Prints and underlines the headings for the chart.
190-250 Does the figuring for 1 to 8 persons:

200 The cost per person (C) is the total price (P), divided by the number
of persons (N).

210 The number of slices per person (SL) is the number of slices in the
pie (8), divided by the number of persons (N).

220 Prints the number of persons. The semicolon holds the cursor at
the present row and waits for the next piece of information to be
printed.

230 Counts to the 8th column in the row and prints the cost per person
(C).

240 Prints the number of slices per person (SL) at the 16th column.
250 Goes back to line 190 to repeat the loop for the next number of

persons.

260 Moves the chart up the screen.
270 A dummy line is used to freeze the chart in position on the screen.

The extra lines (lines 101, 201, and 211) are used to keep you from getting
long decimal numbers in the chart. Lines 101 and 201 round off the amount of
the tip and the cost per person to the nearest hundredth. This formula is often
used to round off money to the nearest penny. Line 211 rounds off the number
of slices to tenths—about as close as you would care to estimate when
dividing a piece of pizza!

Pizza

After the game, you and some friends
go out for pizza. It's hard to decide
what to get. How much will it cost?
How much is the tip? How much
should each person chip in?
How much may each person eat?
Let the computer do your arithmetic.

10 CALL CLEAR
20 PRINT "WHAT KIND DO YOU WANT?"

30 INPUT K$
40 PRINT "HOW MUCH DOES IT COST";
50 INPUT P
60 CALL CLEAR
70 PRINT TAB(5);K$;" PIZZA"
80 PRINT TAB(3);"

MENU

Plain.

Extra Cheese-

Sausage
Pepperoni
Meatball

Peppers
Onions
TheWorks

All pies: 8 slices

.6.95

.7.95

.8.95

.8.95

.8.95
J7.95
.7.95
.9.95

<&> Pitcher of Soda
with Every Pizza Order^

90 PRINT "COST: ";P;

100 TP=.15*P
110 PRINT TAB(14);"TIP: ";TP
120 P=P+TP

130 PRINT " TOTAL COST: ";P

140 PRINT

150 PRINT "NO.OF";

160 PRINT TAB(9);"C0ST/"," SLICES/"
170 PRINT "PERSONS PERSON PERSON"

180 PRINT "

190 FOR N=l TO 8

200 C=P/N
210 SL=8/N
220 PRINT N;
230 PRINT TAB(8);C;
240 PRINT TAB(16);SL
250 NEXT N

260 PRINT ::

270 GOTO 270

101 TP= INT(100*TP+.5)/100
201 C=INT(100*C+.5)/100
211 SL = INT(lO*SL+.5)/10

Main ideas: Keeping a running total during a game of chance

5-20 Gets the game started.
25 A remark to remind you what this section is about.

30-35 Clears the screen and makes it blue (color 5).
40 Because there will be writingon the screen, the boxes will be drawn

using a character not used for printing messages. I chose small A
(character 97 in set 9). This line sets all characters in set 9 to red on
red, or red squares.

45-65 Draws the 3 boxes at rows 7 to 12:

50 The first box starts at column 5 and extends for 5 boxes.

55 The second box starts at column 13 and extends for 5 boxes.

60 The third box starts at column 21 and extends for 5 boxes.

During the game, the total (T) is decreased by 250 each time you play. It is
increased by 500 each time you win a game.

100 Another remark.

105 Randomizes the numbers.

110 Picks a number (M) from 1 to 3, where the money is hidden.
115 Subtracts 250 from your total so you can play.
120 Prints how much money you have left (your new total).
125 Ifyou have 0 or less, you don't have enough money to play—you

skip to line 300. (Bye!)
130 Asks you to choose a box.
135 Allows you to input your guess (G).
140 Sends you to line 200 if your guess was correct.
145 Goes back to line 25 to give you a chance to lose more money.
200 A remark.

201 Adds 500 to your total.
205-225 A repeating loop to make flashes:

210 Prints a blank line, which moves the boxes up the screen.
215 Picks a number from 0 to 13, and adds 3 to that number. (This

effectively eliminates colors 1 and 2, whichwould make it impossi
ble to read messages on the screen.)

220 Makes the screen the color chosen in line 215.

221 Prints your new total.
225 Goes back to line 205 for the next flash.
230 When the flashing is finished, goes back to line 25 to start again.

300-310 Ends the game.

Double or Nothing
"Double or Nothing" is your favorite game at the school fair.
The person at the booth has three little boxes. A half dollar is
hiding in one of them. Guess which one.

5 CALL CLEAR

6 PRINT "HOW MUCH MONEY DO YOU HAVE?"

7 INPUT T

10 PRINT "YOU MAY CHOOSE BOX 1,2 OR 3."
15 PRINT "ARE YOU READY?"

20 INPUT R$
25 REM DRAW BOXES

30 CALL CLEAR

35 CALL SCREEN(5)
40 CALL C0L0R(9,9,9)
45 FOR R=7 TO 12

50 CALL HCHAR(R,5,97,5)
55 CALL HCHAR(R,13,97,5)
60 CALL HCHAR(R,21,97,5)
65 NEXT R

100 REM GAME

105 RANDOMIZE
110 M=INT(3*RND)+1
115 T=T-.25

120 PRINT "YOU NOW HAVE $";T
125 IF T<=0 THEN 300
130 PRINT "WHICH DO YOU CHOOSE";
135 INPUT G

140 IF G=M THEN 200
145 GOTO 25

200 REM YOU WON
201 T=T+.50
205 FOR F=l TO 10
210 PRINT

215 C= INT(14*RND)+3

220 CALL SCREEN(C)
221 PRINT TAB(5);"YOU'RE UP TO $";T
225 NEXT F

230 GOTO 25

300 REM END OF GAME

305 CALL CLEAR
310 PRINT TAB(4);"SORRY YOU ARE BROKE!!"

You pay 250 to play. If you pick
the right box, you keep the half

dollar. You may play until
you go broke*

Main ideas: Using READ, DATA, and RESTORE
Incrementing a total

This program finds the number of days between any twodates that you select.
It does so by converting each date to its numerical position in the year. For
example, January 1 is day 1 in the year, January 31 is day 31, February 1 is
day 32, and December 31 is day 365. The program converts each date by
adding up the number of days ineach monthprecedinga given date and then
adding on the number of days in the given month. To find out what day April
10th is, we reason like this:

April 10 or 4/10 = 31 + 28 + 31 + 10 = 100

days in the 3 months -t t- days in April
preceding April

If the month is January (which is the firstmonth), you don't have to add up any
months. Just use the number of the day. For example, January 13 is day 13.
Now let's look at the program lines individually:

10 Clears the screen.

20-30 Initially, sets day 1 and day 2 at 0.
40-80 Allows you to input the beginning date—month and day. (Enter

numbers for the months—1 for January, 2 for February, etc.)
90 If the month is January, you skip to line 150.

100 Counts to the number just before the given month. (If the month is
5, for example, it counts to 4; that is, ifthe month is May, itwill add
up the number of days in the first 4 months.)

110 Reads the number of days in each month (data line 500).
120 Adds the number of days to day 1.
130 Finishes the loop begun in line 100.
140 Restores the data, which will be read again to figure day 2.
150 Adds the number of the day in the current month to the total days in

the previous months.
160 Prints the number of day 1.

170-290 Figures the number of day 2.

300 Prints 2 blank lines.

310 Finds the number of days between day 1 and day 2.
320-340 Fixes things up when you go over the first day of the year. For

instance, without this line, the computer would say that there are
-364 days from December 31 to January 1!

350 Prints the results.

500 The data line that holds the number of days in each month.

Days to Go
No matter how much fun school is, everyone likes to figure out
how many days until summer vacation. While you're at it,
how many days until your birthday? Until Christmas?
Until Halloween? Until whenever??

10 CALL CLEAR

20 D1=0
30 D2=0

40 PRINT "HOW MANY DAYS"

50 PRINT "FROM MONTH";
60 INPUT M

70 PRINT " DAY";
80 INPUT D

90 IF M=l THEN 150
100 FOR X=l TO M-l

110 READ N

120 LET D1=D1+N

130 NEXT X

140 RESTORE

150 LET D1=D1+D
160 PRINT "DAY 1 IS #

170 PRINT :

180 PRINT "TO MONTH";

190 INPUT M
200 PRINT " DAY";
210 INPUT D

220 IF M=l THEN 280
230 FOR X=l TO M-l

240 READ N

250 LET D2=D2+N

260 NEXT X

270 RESTORE

280 LET D2=D2+D
290 PRINT "DAY 2 IS #";D2

300 PRINT :

310 T=D2-D1

320 IF T<0 THEN 340
330 GOTO 350

340 T=T+365

350 PRINT "THERE ARE";T;"DAYS TO GO"

500 DATA 31,28,31,30,31,30,31,31,30,31,30,31

APPENDIX

Character Sets, 45
Colors, 47
Graphics Sheet, 48-49

